Oppgave 10

Parametre i porteføljе av aksjer og obligasjoner

Porteføljе av aksjer og obligasjoner fra oppgave 2:
$\kappa X_t + (1 - \kappa) Y_t$

```
μ = 0.10;
ν = 0.05;
σ = 0.20;
τ = 0.05;
ρ = 0.40;

sigma[κ_] := √(κ^2 σ^2 + (1 - κ)^2 τ^2 + 2 ρ (1 - κ) τ κ σ);
my[κ_] := κ μ + (1 - κ) ν;

dsRound[tall_, des_] := N[Round[10^des tall] / 10^des];

TableForm[Table[{κ, dsRound[my[κ], 3], dsRound[sigma[κ], 3]}, {κ, 0, 1, .1}],
 TableHeadings → {None, {"κ", "my", "sigma"}}]
```

κ	my	sigma
0	0.05	0.05
0.1	0.055	0.056
0.2	0.06	0.067
0.3	0.065	0.081
0.4	0.07	0.096
0.5	0.075	0.112
0.6	0.08	0.129
0.7	0.085	0.147
0.8	0.09	0.164
0.9	0.095	0.182
1.	0.1	0.2

Tester modellen med garanti mot modellen uten garanti

Parametre:
gG = årlig regulering av folketrygdens grunnbeløp, G
gL = årlig regulering av lønn

```
x = 30;
ω = 120;
n = 67 - x;
m = ω - 67;
lønn = 400000;
grunnbeløp = 56861;
gL = 0.04;
gG = 0.03;
```

```
<< "Statistics`ContinuousDistributions`"
<< "Graphics`Legend`"
<< "Graphics`Graphics`"
```

nSim, a, l, l67, g, p, ft er som i oppgave 9
info = gir ytelse fra innskuddspensjon som en vektor {uten garanti, med garanti}

```
nSim = 10000;

aSim[κ_] := Partition[e^(my[κ] - sigma[κ]^2/2 + sigma[κ] RandomArray[NormalDistribution[0,1], nSim n]), n];

Timing[a = aSim[0.3];]
```

```
{0.761 Second, Null}
```

```
l = Table[lønn (1 + gL)^i, {i, 0, n}];
l67 = Table[l[[67 - x + 1]] (1 + gG)^i, {i, 0, m}];
g = Table[grunnbeløp (1 + gG)^i, {i, 0, n}];
p = Table[0.05 Min[l[[i + 1]] - 2 g[[i + 1]], 4 g[[i + 1]]] UnitStep[l[[i + 1]] - 2 g[[i + 1]]] +
    0.08 Min[l[[i + 1]] - 6 g[[i + 1]], 6 g[[i + 1]]] UnitStep[l[[i + 1]] - 6 g[[i + 1]]], {i, 0, n - 1}];
ft = 0.80 g[[n + 1]] + 0.42 Min[l[[n + 1]] - g[[n + 1]], 5 g[[n + 1]]] UnitStep[l[[n + 1]] - g[[n + 1]]] +
   (0.42 / 3) Min[l[[n + 1]] - 6 g[[n + 1]], 6 g[[n + 1]]] UnitStep[l[[n + 1]] - 6 g[[n + 1]]];

info[ξ_, sim_] :=
  Module[{konto}, konto = FoldList[(#1 + #2[[2]]) * #2[[1]] &, 0, Transpose[{a[[sim]], p}]];
   kontoMedGaranti = FoldList[(#1 + #2[[2]]) * (1.03 + ξ Max[0, #2[[1]] - 1.03]) &,
     0, Transpose[{a[[sim]], p}]]; (1/14 {konto[[n + 1]], kontoMedGaranti[[n + 1]]}) / l[[n + 1]]];

folketrygdPensjon = ft / l[[n + 1]]
```

```
0.344817
```

```
Timing[livrente = Table[info[0.5, i], {i, 1, nSim}];][[1]]
```

```
11.887 Second
```

```
forventetPensjon = Mean[livrente]
```

```
{0.202894, 0.169155}
```

```
Show[Histogram[folketrygdPensjon + Transpose[livrente][[1]],
   BarStyle → RGBColor[1, 0, 0], DefaultFont → {"Verdana", 13},
   HistogramRange → {0, 1}, DisplayFunction → Identity],
  Histogram[folketrygdPensjon + Transpose[livrente][[2]], BarStyle → RGBColor[1, 1, 1],
   DefaultFont → {"Verdana", 13}, HistogramRange → {0, 1}, DisplayFunction → Identity],
  Graphics[{AbsolutePointSize[7], Point[{folketrygdPensjon, 0}]}],
  Graphics[Text["Folketrygd", {folketrygdPensjon, nSim/100}]],
  DisplayFunction → $DisplayFunction, ImageSize → 600];
```



3D plott

```
nSim = 1000;
```

```
aSim[κ_] := Partition[e^(my[κ] - sigma[κ]^2/2 + sigma[κ] RandomArray[NormalDistribution[0,1],nSim n]), n];
```

Sannsynlighet for tap for forsikringsselskapet er lik sannsynligheten for at kontoen med garanti er høyere enn kontoen uten garanti:

```
prTap[ξ_, κ_] := Module[{a, konto, kontoMedGaranti, diff = Table[0, {nSim}]},
   a = aSim[κ]; Do[konto = FoldList[(#1 + #2[[2]]) * #2[[1]] &, 0, Transpose[{a[[sim]], p}]];
    kontoMedGaranti = FoldList[(#1 + #2[[2]]) * (1.03 + ξ Max[0, #2[[1]] - 1.03]) &, 0,
      Transpose[{a[[sim]], p}]]; diff[[sim]] = kontoMedGaranti[[n + 1]] - konto[[n + 1]],
    {sim, 1, nSim}]; N[Length[Select[diff, #1 > 0 &]]/nSim]];
```

```
Timing[tb = Table[prTap[ξ0, κ0], {ξ0, 0, 1, .2}, {κ0, 0, 1, .2}];]
```

```
{42.802 Second, Null}
```

```
tbInt = ListInterpolation[tb, {{.1, .9}, {.1, .9}}];
```

```
lp = Plot3D[tbInt[x, y], {x, .1, .9},
   {y, .1, .9}, AxesLabel → {"ξ", "κ", "ϕ"}, ImageSize → 350];
```




```
ContourPlot[tbInt[x, y], {x, .1, .9}, {y, .1, .9},
  Contours → Table[i, {i, 0, 1, .1}], FrameLabel → {"ξ", "κ"}];
```




```
<< "JavaView`JLink`"
InstallJavaView[];
```

```
JavaView CodeBase = C:/Program Files/
  Wolfram Research/Mathematica/5.0/AddOns/Applications/JavaView/
```

```
JavaView[lp];
```

```
geom = JavaView[lp];
geom[showTransparency[True]];
geom[setTransparency[0.4]];
geom[update[geom]]
```

```
True
```

```
UninstallJavaView[];
```